

Copyright (C) 2021 Timothy Fossum

Permission is granted to copy, distribute and/or modify this document under the terms of the GNU Free Documentation License, Version 1.1 or any later version published by the Free Software Foundation. A copy of the license is included in the file "COPYING", entitled "GNU Free Documentation License".

Classes and Objects

Compared to most object-oriented programming languages such as C++, our classes and objects are *first class* – that is, they are values that can be used at any time and in any environment, they can be bound to variables, they can be passed as arguments to procedures, and they can be returned as values from applications.

A *class* is an expressed value that captures a collection of variables (called *slots*) and procedures (called *methods*) and that serves as a template to create new values called *instances* of the class. The instances are called *objects* and are distinguished from classes. You can think of a class as a factory that identifies and creates out an arbitrary number of instances (objects) of the class.

Language OBJ

All classes of an OBJ program belong to a *class hierarchy*, which is a tree with an unnamed class at the root of the tree and with program-created classes as the other nodes of the tree.

In the class hierarchy, a class X that occurs as a child node of class Y in the hierarchy is called a *subclass* of Y , and Y is called a *superclass* of X ; we also say that X *extends* Y .

When an object of class X is instantiated, instances of each of the classes on the path from X to the root of the tree are created, and the combination of these instances is considered as the resulting object.

If Y is the superclass of X , then an object x created from class X “contains” an object y created from Y . The object y is called the *parent* of x , and likewise x is called the *child* of y .

A class may also have `static` variables whose values are shared by all instances of the class.

If x is an object and f is a field, the expression $\langle x \rangle f$ evaluates to the value of f in object x . In languages like C++ and Java, this would be written as $x.f$.

Language OBJ (continued)

Consider the following example.

```
define c1 =
  class % extends the unnamed top-level class
    field x
    field y
  end

define c2 =
  class extends c1
    field z
  end

define o2 = new c2

<<o2>super>set x = 3
<o2>x % evaluates to 3
```

In this example, `c1` is a class and `c2` is a subclass of `c1`. Object `o2` is an instance of `c2` and `<o2>super` is an instance of `c1`.

Language OBJ (continued)

We define the concrete and abstract syntax of classes and objects the additional reserved word `nil` to our language, which represents a value not shared by any other data type. When used in boolean expressions, `nil` is considered false. (The only other such value is 1 zero.)

```
<exp>:NilExp      ::= _NIL
                  NilExp ()
<exp>:ClassExp    ::= _CLASS_<ext>_<statics>_<fields>_<methods>_E1
                  ClassExp (Ext_ext, _Statics_statics, _Fields_fiel
<ext>:Ext0        ::= _
                  Ext0 ()
<ext>:Ext1        ::= _EXTENDS_<exp>
                  Ext1 (Exp_exp)
<statics>        **=_STATIC_<VAR>_EQUALS_<exp>
                  Statics (List<Token>_varList, _List<Exp>_expList
<fields>         **=_FIELD_<VAR>
                  Fields (List<Token>_varList)
<methods>        **=_METHOD_<VAR>_EQUALS_<proc>
                  Methods (List<Token>_varList, _List<Proc>_procLi
<exp>:NewExp      ::= _NEW_<exp>
                  NewExp (Exp_exp)
<exp>:EnvExp      ::= _LANGLE_<exp>_vExp_RANGLE_<exp>_eExp
                  EnvExp (Exp_vExp, _Exp_eExp)
```

Language OBJ (continued)

Every `Exp` expression evaluates to a `Val` object, and a `ClassExp` expression, so evaluating such a `ClassExp` (*i.e.*, calling its `eval` method) returns a `ClassVal` object. Looking only at the syntax, it seems reasonable that such a `ClassExp` consists of (possibly empty) superclass (the `ext` [for EXTENDS] part), a list of variables bound to values, a list of field names, and a list of method names and their procedures.

We define a `StdClass` object in the file `class`, which extends `ClassVal`. (Actually, it extends `ClassVal` class which in turn extends the `Val` class.) Evaluating a `ClassExp` expression returns an instance of `StdClass`.

[Warning: In this chapter we implement classes and objects in our source language (OBJ), using classes and objects in our implementation language (J). This can be confusing. For example, a `ClassExp` evaluates to an object in J, but represents a class in our source language. Be sure that you have a clear understanding of the context in which the terms “class” and “object” are being used in this discussion.]

Language OBJ (continued)

A `StdClass` object (we're in the implementation language, Java) has the following instance variables:

```
public ClassVal superClass;  
public Fields fields;  
public Methods methods;  
public Bindings staticBindings;  
public Env staticEnv;
```

Language OBJ (continued)

The `superClass` variable is a reference to a `ClassVal` object representing the superclass of this class. The `staticBindings` variable is a reference to a `Bindings` object containing the bindings of the static variable names to their RHS values. The `staticEnv` variable is a reference to an `Env` object evaluated in the current static environment. (More about this later.)

The `fields` and `methods` variables are references to the `Fields` and `Methods` objects that capture the class field names and method names and pointers to the corresponding objects. These objects are not evaluated yet.

The `staticEnv` variable is a reference to the static environment of the class. It starts out extending the static environment of the superclass by adding new `Bindings` object named `staticBindings`. New bindings are added to the `staticBindings` list using the variable definitions in the `staticFields` parameter of the `StdClass` constructor. Each static LHS identifier is mapped to the value of its RHS, where the RHS expression is evaluated in the current static environment. These bindings are created in order (first to last) as in top-level

Language OBJ (continued)

Two predefined identifiers are inserted initially into the list of static bindings. The identifier `myclass` is bound to (a reference to) this class itself, and the identifier `superclass` is bound to (a reference to) the superclass of this class.

If the class expression does not have an `extends` part, its superclass is the unnamed “parentless” class (a static `EnvClass` Java object) whose static environment is the top-level program environment – so top-level variables defined are visible. In this way, all of the RHS expressions in the static definitions of such a class have access to the other static bindings in the environment as well as to the top-level bindings.

A class expression that specifies an explicit superclass (using `extends`) has access to the static environment of that superclass, and its superclass chain – has access to the bindings in all of the superclass environments as well as to the top-level bindings.

If a class expression has a static definition for a variable that also appears in a superclass, that definition shadows the superclass variable. We disallow duplicate LHS variable names in a given class expression’s static definitions. We also disallow static redefinitions for `myclass` or `superclass`.

Language OBJ (continued)

Although counter-intuitive, objects are actually simpler than classes. An object is essentially a wrapper for an instance of `Env`!

An `ObjectVal` is a Java class that extends the `Val` class. It has a static field `env` of type `Env`:

```
public Env objectEnv;
```

The `new` operator in our source language takes a class expression and a static environment, and returns a Java `ObjectVal` instance that essentially couples the static bindings of the class with bindings for the class fields and methods.

Since our language does not define an explicit constructor in class expressions, we initially bind object fields to (a reference to) `nil`.

The method variable names in the class definition are bound to procedure objects that capture the environment that includes bindings for the class variables (from the `staticBindings` field), along with bindings for the fields described above. The method closures are created as in `letrec`, so they point to themselves recursively. As with static and field definitions, we disallow duplicate method variable names.

Language OBJ (continued)

Before we build an object from a base class, we first build an object that is an instance of the superclass of the base class. This superclass object has an environment, namely `objectEnv`. We then extend this superclass object's environment by adding bindings to the statics, fields, and methods of the class, and then use the extended environment to create the instance of the base class.

Since creating the superclass object may itself involve creating an object that is an instance of its superclass, object creation continues up the class hierarchy until an `EnvClass` class is found, at which point there is no further superclass to create.

At the top of the chain of superclass objects, we add the identifier `self` to the environment of this top-level superclass object, binding it to a reference to the base class object being created. Since all of the objects created by going up the superclass chain have environments that ultimately extend the top-level environment, these objects can refer to the base class object being created using the `self` identifier. (In Java, we do the same thing using `this` instead of `self`.) Statics declared in superclasses that refer to `self` will “see” the base class object. This is important for dynamic dispatch of method calls, an important feature of object-oriented languages. We call this binding of `self` a *deep* binding.

Language OBJ (continued)

As each object is created up the superclass chain, we insert three bindings into its list of field bindings: `this`, `self`, and `super`. We bind the identifier `self` to the base object being created, which is the same as describing the object being created. We bind the field identifier `this` to the object being created at the particular point in the superclass chain (we call this a *shallow* binding). And we bind the identifier `super` to the superclass object. The code for creating these bindings is in Slide 5.16.

We disallow duplicate field names in class definitions. We also disallow field names that duplicate the predefined identifiers `this`, `self`, and `super`.

Notice that an object can see all of the static bindings up the superclass chain, but that if a static variable is bound to a procedure (or some other object that captures an environment), the procedure captures only the static environment of the class and cannot “see” any of the fields or methods – including the `self` identifier – in its environment.

Language OBJ (continued)

An instance of the Java `EnvClass` class has one instance variable:

```
public Env staticEnv;
```

Here is the code for the `EnvClass` constructor:

```
public EnvClass(Env env) {  
    // the static environment of this class  
    staticEnv = env;  
}
```

The `EnvClass` defines a static `envClass` Java object that is a instance of an `EnvClass`. This `envClass` object is the root of the tree whose `staticEnv` is the top-level environment.

A standard class (an instance of the Java `StdClass` class) has a constructor, except that it builds on the environment of its superclass as defined. A standard class defines bindings for the variables `myclass` and `myfile` whose values are self-explanatory. You can find the code for this `class` file `class`.

Language OBJ (continued)

As described on slides 9 and 10, as an object is created, the selector is bound to the `objRef` reference that is passed to the `makeObject` method of the `EnvClass`. But how can this binding take place when the object has not been completely created? We do this by creating a Java `ValRef` object named `objRef` with a dummy initial value (`nil`, to be precise), and pass this to the `makeObject` method.

The `objRef` is passed up the superclass chain through successive `makeObject` calls. When `makeObject` reaches the `EnvClass` Java class, the selector is put into the object environment, bound to `objRef`. After the original object has been completely created – that is, after all of the `makeObject` calls have returned – `objRef` is rebound to the newly created object with a call to `setRef`. The complete `eval` code for a new expression in the `NewExp` part of the

```
public Val eval(Env env) {  
    // get the class from which this object is created  
    Val val = exp.eval(env);  
    // create a reference to a dummy value (nil)  
    ValRef objRef = new ValRef(Val.nil);  
    // let the class create the object  
    ObjectVal objVal = val.makeObject(objRef);  
    // set the reference to the newly created object  
    return objRef.setRef(objVal);  
}
```

Language OBJ (continued)

The `makeObject` method is defined for both a `StdClass` and a `Class`.
(For all other `Val` objects, `makeObject` throws an exception.)

In `StdClass`, `makeObject` first creates an instance of the superclass, then it stitches together an environment that includes the static bindings, the instance bindings, and the methods. Three fields are created and initialized automatically: `super` is bound to the superclass object (a deep binding), `super` is bound to the instance of the superclass (a shallow binding), and `this` is bound to this object (a shallow binding). The remaining bindings are bound to `nil`, and the methods are bound to closures as in `letrec`. The definition for `makeObject` is given on the next two slides.

Language OBJ (continued)

```
public ObjectVal makeObject(Ref objRef) {
    // create the parent object first (recursively)
    ObjectVal parent = superClass.makeObject(objRef);

    // this object's environment extends the parent's
    Env env = parent.objectEnv;

    // add this class's static bindings
    env = env.extendEnvRef(staticBindings);

    // the fields come next
    Bindings fieldBindings = new Bindings();
    env = env.extendEnvRef(fieldBindings);
    // bind all of this object's instance fields to
    fields.addFieldBindings(fieldBindings);

    // bind all this object's methods as in letrec
    env = methods.addMethodBindings(env);
}
```

... continued on next slide ...

Language OBJ (continued)

... continued from previous slide ...

```
// create the object
ObjectVal objectVal = new ObjectVal(env);

// bind 'super' field to the parent object
fieldBindings.add("super", new ValRef(parent));
// bind 'self' field to the base object being created
// (to speed up lookups)
fieldBindings.add("self", objRef); // deep
// bind 'this' field to this object environment
fieldBindings.add("this", new ValRef(objectVal));
return objectVal;
}
```

Observe that this code binds `self` to `objRef` in every set of field bindings created recursively up to the top-level `EnvClass` class. This is not necessary because the top-level `EnvClass` object is guaranteed to have a field binding for `self` (a reference to `objRef`), so any reference to `self` will eventually be found in the chain of environments. However, by putting the binding in every intermediate object environment, a reference to `self` will be found sooner in `applyEnv` lookups.

Language OBJ (continued)


The `makeObject` method for `EnvClass` is simple, since it's always creating the same superclass object that needs to be constructed. It extends the `EnvClass` object (namely the top-level environment, with a single field binding of `self` (a reference to) the object being created. This binding is “deep”, in that the `objRef` reference may ultimately refer to an object defined in a subclass. As shown in the `NewExp` code for `eval` on Slide 5.13, objects are created in the chain of superclasses, `objRef` is finally bound to the object at the beginning of the chain.

```
public ObjectVal makeObject(Ref objRef) {
    // start with the static (top-level) env. of the program
    Env env = staticEnv;
    // add the field binding 'self' to refer to
    // the object being created (objRef)
    Bindings fieldBindings = new Bindings();
    fieldBindings.add("self", objRef);
    env = env.extendEnvRef(fieldBindings);
    return new ObjectVal(env);
}
```

Observe that an `ObjectVal` can access the static environment of the `EnvClass` class, which is the top-level program environment.

Language OBJ (continued)


Here is a diagram showing the components of a `ClassVal`:



On the next six slides, we show step-by-step how to create an instance using the `new` operator.


Language OBJ (continued)

Here is the class we want to create an instance of:




Language OBJ (continued)

Step 1: create a dummy reference `objRef` to a `NilVal`




Language OBJ (continued)

Step 2: make a parent object (recursively), binding `self` to the object




Language OBJ (continued)

Step 3: Extend the parent object environment by the static bindings



Language OBJ (continued)


Step 4: Extend the environment of Step 3 with new field bindings



Fields are initialized to **nil**. Add bindings for **this**, **self**, and **super**


Language OBJ (continued)

Step 5: Extend the object environment with method bindings, as in



Language OBJ (continued)


Step 6: Set `objRef` to refer to the newly created object



The `ObjectVal` instance is the value returned by the new operator

Language OBJ (continued)

This diagram shows both the object environments and the static env



Observe that the class static bindings belong to both environments.

Language OBJ (continued)

The RHS expressions in static definitions are evaluated in the static environment of the class, which extends the static environment of its superclass (and so on). For example, consider the following definitions and expression:

```
define c = class static x=5 end
define x = 3
<class extends c static y=x end>y
```

In the class that extends *c*, the value of *x* on the RHS of the static definition can be found in the static environment of the superclass *c*, which is bound to 5. Thus the variable *y* is bound to 5, not 3, so the expression evaluates to 5.

Now consider a variant of the expression:

```
define c = class static xx=5 end
define x = 3
<class extends c static y=x end>y
```

In the class that extends *c*, the value of *x* on the RHS of the static definition can be found in the static environment of superclass *c*. But *c* is not a variable *x*, so the value of *x* must be found in the static environment of the top-level environment. In the top-level environment, the variable *x* is bound to 3, so the expression evaluates to 3.

Language OBJ (continued)

The $\langle \dots \rangle$ operator, when applied to a class, extracts the static environment of the class. When used in an expression of the form

$\langle \dots \rangle \text{exp}$

the expression exp is evaluated in the static environment of the given class. This can be used to evaluate an expression that refers to any static variable.

Consider a class c , for example. For a static variable x in the class c , the expression $\langle c \rangle x$ evaluates to the value of the variable. For a static procedure f in class c , the expression $\langle c \rangle f(\dots)$ evaluates to the application of f to the given parameters.

When making a procedure application such as $\langle c \rangle f(\dots)$, it is important to know

- the environment in which f is evaluated, and
- the environment in which the actual parameters to f are evaluated.

Language OBJ (continued)

To evaluate the expression $\langle c \rangle f(\dots)$, we evaluate the procedure f in the environment determined by $\langle c \rangle$, whereas we evaluate the actual parameters in the environment in which the application $\langle c \rangle f(\dots)$ is made. For example, in

```
define f = proc(t) *(2,t)
define c =
  class
    static x = 3
    static f = proc(t) t
  end
let
  x = 5
in
  .<c>f(x)
```

the expression $\langle c \rangle f(x)$ evaluates to 5, because f is bound to the procedure given in the static definition of c , but its actual parameter x is evaluated in the local `let` environment. In other words, $\langle c \rangle f(x)$ is the same as $\{ \langle c \rangle f \} (x)$.

Language OBJ (continued)

In particular, observe that the following two expressions are not equivalent:

```
.<c>f ( . . . )  
<c>.f ( . . . )
```

In the first expression, `f` is evaluated in the static environment of `c`. The procedure bound to `f` is applied to the actual parameters `(. . .)` which are in the current environment, *not* in the static environment of `c`.

In the second expression, the entire expression `.f (. . .)` is evaluated in the environment of `c`, which means that the actual parameters `(. . .)` are evaluated in this static environment.

In the example on the previous slide, if the final expression was `<c>` of `.<c>f (x)`, it would evaluate to 3, since `x` is bound to 3 in the environment of `c`.

To clarify, the above two expressions can be re-written as follows:

```
. { <c>f } ( . . . )  
<c> { .f ( . . . ) }
```

Language OBJ (continued)

The static environment of a class ultimately ends up extending the program environment, not the environment in which the class is defined. The following code:

```
define x = 3  
let  
  x = 5  
in  
  <class end>x
```

In this example, the class is defined in the `let` environment, but its static environment extends the top-level environment, not the `let` environment, so this expression is 3, not 5.

There are situations in which we may want to retrieve the value of a “local” environment in which the class is defined and not in the static environment of the class. To do so, we predefine a static “variable” `!@` (called `!@`) in every class and bind it to an object that captures the (local) environment in which the class is defined. This binding becomes part of the static environment of the class. The token `!@` is not really a variable, so it cannot appear in an assignment. Its principal use is in expressions of the form

```
<!@>exp
```

which evaluates to the value of the expression `exp` in the local environment.

Language OBJ (continued)

For example, consider again the expression

```
define x = 3
let
  x = 5
in
  <class end>x
```

where we observed that this expression evaluates to 3. If we replace

```
<class end>x
```

in the above expression with

```
<class end><!@>x
```

then this expression evaluates to 5 because the `class` expression is a `let` environment, and so the environment captured by `!@` has `x` bound to 5. The above expression can also be written as follows:

```
<<class end>!@>x
```

Observe that the “variable” `!@` is only defined in the static environment and is bound to an object that captures the environment in which the `class` expression is evaluated. If used anywhere else, it will throw an “unbound variable” exception.

Language OBJ (continued)

The special operator `@` returns an object that captures the *current* environment, whatever that may be. (Recall your homework assignment that introduced this operator.) This operator may be used to pass a captured environment to a procedure application or to assign it to a variable for later reference.

```
AtExp
%%%
    public Val eval(Env env) {
        return new ObjectVal(env);
    }

    %%%
```

The special operator `@@` does the same thing, except that it also displays the current environment in a human-readable way.

Notice that `@` is meaningful in *any* expression context – because every expression is evaluated in *some* environment – but that `!@` is only meaningful in the context of a `class` expression, and its value represents an environment that is entirely separate from the static environment of the class.

Language OBJ (continued)

Some examples:

```
define x = 11
define y = 42
define z = 666
define xyenv =
  let
    x = 3
    y = 5
  in
    @
<@>x      % => 11
<@>y      % => 42
<@>z      % => 666
<xyenv>x  % => 3
<xyenv>y  % => 5
<xyenv>z  % => 666 (the 'let' extends the tc
```

Language OBJ (continued)

Observe that for any expression `exp`, the following two expressions have the same values:

```
<@>exp
exp
```

Language OBJ (continued)

Unlike the `new` operator in Java, Our `new` operator does not take arguments and all of the fields are initialized to `nil`. We can, of course, initialize by calling a method. Here's an example:

```
let
  c = class
    field x
    field y
    method init = proc(a,b) { set x=a ; set y=b }
  end
in
  let
    o = .<new c>init(3,4)
  in
    <o>+(x,y) % => 7
```

Since the `init` method returns `self`, the value of `t.<new_c>init(3,4)` is the *same* object as the one created by `new_c`, except that its fields `x` and `y` are set to values 3 and 4, respectively.

You might be inclined to think that `.<new_c>init(a,b)` is equivalent to `<new_c>{_set_x=a_;_set_y=b_;_self}`, but the binding of `self` may be different in these two expressions.

Language OBJ (continued)

In the previous example, the `init` method is invoked separately, and the object is created using `new`, and not as part of the object creation itself. The procedure “`init`” is not a requirement. A class can have several methods to initialize its fields, much as a Java class can have several constructors. In Java, the OBJ language can apply its methods – even the ones intended to initialize the fields – at any time.

```
define c =
  class
    field x
    method init = proc() {set x = 5 ; set y = 4}
    method foo = proc() {set x = add1(x)}
  end
```

```
define o = .<new c>init()
<o>x % => 5
<o>{.foo() ; x} % => 6
<o>{.init() ; x } % => 5
```

Language OBJ (continued)

The OBJ language has three additional expressions, with the following rules:

```
<exp>:DisplayExp ::= _DISPLAY_<exp>
                               DisplayExp(Exp_exp)
<exp>:Display1Exp ::= _DISPLAY1_<exp>
                               Display1Exp(Exp_exp)
<exp>:NewlineExp ::= _NEWLINE
                               NewlineExp()
```

The DISPLAY, DISPLAY1, and NEWLINE tokens are defined by

```
DISPLAY 'display'
DISPLAY1 'display#'
NEWLINE 'newline'
```

Evaluating a DisplayExp expression results in evaluation of its sub-expression in the current environment; this value's toString() representation is displayed on standard output, and the value is returned as the value of the DisplayExp expression. Display1Exp is like DisplayExp except that the displayed value is followed by a single space. The value of a NewlineExp expression is nil, and a newline is displayed on standard output.

Language OBJ (continued)

Here are examples of how to use display, display#, and newline:

```
let
  x = 3
  y = 5
  z = 8
in
  { display x ; newline
    ; display y ; newline
    ; display z ; newline
    ; nil
  }
```

Evaluating this expression results in displaying the following to standard output (nil):

```
3
5
8
```

If the newline expressions are removed, the output appears as follows (omitting nil):

```
358
```

If display is then replaced by display#, the output appears as follows:

```
3 5 8
```

Language OBJ (continued)

Consider the following OBJ program:

```
define summer =
  class
    field sum
    method init = proc() {set sum = 0 ; self}
    method add = proc(t) {set sum = +(sum,t) ; self}
    method show = proc() {display sum ; newline ; self}
  end
```

Here's an example of how this class might be used to find and display the integers 1, 3, 5, and 7:

```
define o = .<new summer>init()
.<o>add(1)
.<o>add(3)
.<o>add(5)
.<o>add(7)
.<o>show()
```

Since `init` and `add` both return `self`, the following “one-liner” accomplishes the same thing:

```
.<.<.<.<.<.<.<new summer>init()>add(1)>add(3)>add(5)>add(7)>show()
```

Language OBJ (continued)

Since we often find ourselves encountering expressions like the following:

```
.<.<.<.<.<.<.<new summer>init()>add(1)>add(3)>add(5)>add(7)>show()
```

we introduce a short-hand way of writing this:

```
!<new summer>init()>add(1)>add(3)>add(5)>add(7)>show()
```

The token `LLANGLE`, defined as the string ‘!<’, introduces an expression that must evaluate to an environment (class or object) – followed by a sequence of one or more procedure applications, each preceded by ‘>’. Each procedure is evaluated in the environment of the previous class or object, and the procedure itself must return another object, whose environment is then used for the next procedure, and so on. The entire expression is terminated by the token, defined as ‘!>’. Note that the actual parameter expressions appear in the environment in which the entire expression appears.

Language OBJ (continued)

Here are the associated grammar rules:

```
<exp>:EenvExp ::= _LLANGLE_<exp>_<mangle>_RRANGLE
EenvExp (Exp_exp, _Mangle_mangle)
<mangle> ::= _RANGLE_<exp>_LPAREN_<rands>_RPARE
Mangle (List<Exp>_expList, _List<Rands>
```

The BNF identifier `mangle` should suggest “multiple angle (bracket)” and could also appropriately be interpreted as a twisted mess.

Language OBJ (continued)

```
<exp>:EenvExp ::= _LLANGLE_<exp>_<mangle>_RRANGLE
EenvExp (Exp_exp, _Mangle_mangle)
<mangle>:Mangle ::= _RANGLE_<exp>_LPAREN_<rands>_RPARE
Mangle (List<Exp>_expList, _List<Rands>
```

Here is the implementation of how to evaluate an `EenvExp` expression:

```
EenvExp
%%%
    public Val eval(Env env) {
        Val v = exp.eval(env);          // the environment
        return mangle.eval(v, env);
    }
%%%
```

The expression `exp` is evaluated in the current environment. Its value, in the current environment, is passed to the `mangle` object, which then handles subsequent procedure applications.

Language OBJ (continued)

```
<exp>:EenvExp ::= _LLANGLE_<exp>_<mangle>_RRANGLE  
                EenvExp(Exp_exp, _Mangle_mangle)  
<mangle>:Mangle ::= _RANGLE_<exp>_LPAREN_<rands>_RPA  
                Mangle(List<Exp>_expList, _List<Ranc
```

A mangle object consists of a list of Exp objects and a list of Ran

Language OBJ (continued)

Here is the code for eval (Val_v, _Env_env) in the Mangle o

```
public Val eval(Val v, Env env) {  
    Iterator<Exp> expIter = expList.iterator();  
    Iterator<Rands> randsIter = randsList.iterator();  
    while (expIter.hasNext()) {  
        // expIter.next() ProcExp to apply  
        // v.env() is the environment in which to b  
        Val p = expIter.next().eval(v.env());  
        // evaluate this method's rands in env  
        List<Val> valList = randsIter.next().evalRan  
        v = p.apply(valList);  
    }  
    return v;  
}
```

Each Exp is evaluated in the environment defined by the value v (which may, of course, be a class or object), and this must evaluate to a ProcVal (which may, of course, be a class or object). The valList arguments to p are evaluated in the outside environment (not in the environment defined by v) from the corresponding randsList. The procedure p is then applied to these arguments, and the result becomes the new v. This is repeated until all of the Mangle a have been performed. The final value v is returned as the result of the EenvExp

Language OBJ (continued)

It turns out that exposing the environment of a procedure can be used with procedures alone, a simplified approach to objects and methods.

OBJ/Prog/pobj

for an example. On the other hand, exposing the environment of a result in modifying that environment, which can lead to unintended

Language OBJ (continued)

In method application, `self` always refers to the base object, even in the definition of a superclass method. This is what makes dynamic dispatch work!

```
define c1 =
  class
    method m1 = proc() 1
    method m2 = proc() <self>m1()
  end
define c2 =
  class extends c1
    method m1 = proc() 2
  end
define o1 = new c1
define o2 = new c2

.<o1>m1() % => 1
.<o2>m1() % => 2
.<o2>m2() % => 2!
```

The following slide gives another example of dynamic dispatch.

Language OBJ (continued)

```
define shape =
  class
    method area = proc() 0 % shapeless
  end

define rectangle =
  class extends shape
    field len % length
    field wid % width
    method init = proc(len,wid) {set <this>len=len ; set <this>wid=wid}
    method area = proc() *(len,wid)
  end

define circle =
  class extends shape
    field rad % radius
    method init = proc(rad) {set <this>rad=rad ; self}
    method area = proc() *(3,*(rad,rad)) % a bit of an underes
  end

define r = .<new rectangle>init(4,5) % a rectangle with length 4 and width 5
define c = .<new circle>init(2)      % a circle with radius 2
define s = new shape

.<r>area() % => 20
.<c>area() % => 12
.<s>area() % => 0
```

Language OBJ (continued)

Other examples on this slide and the next ...

```
define c1 =
  class
    method m1 = proc() <self>m2()
    method m2 = proc() 13
  end
define c2 =
  class extends c1
    method m1 = proc() 22
    method m2 = proc() 23
    method m3 = proc() <super>m1()
  end
define c3 =
  class extends c2
    method m1 = proc() 32
    method m2 = proc() 33
  end
define o3 = new c3

<o3>m3() % => 33
```

Language OBJ (continued)

```
define a = class
  field i field j
  method setup = proc() {set i=15; set j=20; 50}
  method f = proc() .<self>g()
  method g = proc() +(i,j)
end
define b = class extends a
  field j field k
  method setup =
    proc() {set j=100; set k=200; .<super>setup(); .<self>h()}
  method g = proc() [i,j,k]
  method h = proc() .<super>g()
end
define c = class extends b
  method g = proc() .<super>h()
  method h = proc() +(j,k)
end
let
  p = proc(o)
  let
    u = .<o>setup()
  in
    [u, .<o>g(), .<o>f()]
in
  [.p(new a), .p(new b), .p(new c)]

% returns [[50,35,35],[35,[15,100,200],[15,100,200]],[300,35,35]]
```

Language PROP

In many object-oriented programming languages, the fields of an object can be made *private* – that is, inaccessible outside of the object's methods. In such languages, special publically accessible methods can be used to retrieve or to modify them. These methods are often called *getters* and *setters*.

Suppose, for example, we provided a special designator called `<private>` that served to identify a field whose value was inaccessible outside of the object. Consider the following code:

```
define c =
  class
    private x
    method get_x = proc() x
    method set_x = proc(v) set x = v
  end
define cc = new c
.<cc>set_x(5) % ok - sets value of x to 5
.<cc>get_x() % ok - returns 5
<cc>x % illegal - x is private
```

While this sort of code is common, there are two problems with this approach. The first is that every private field we want to access needs a getter and a setter. The second is that code such as `<cc>set_x_=5` does not work, but is intuitive. We need to understand that `.<cc>set_x(5)`.

Language PROP (continued)

The C# language championed by Microsoft solves these problems using a *property*. A property acts like a field but it provides built-in getter and setter code. When the field is *accessed*, the getter code is executed; when it is *assigned to* with a `set` statement, the setter code is executed.

Here is the same class as described on the previous slide, with a property `x` and a getter and setter:

```
define c =  
  class  
    field x  
    property x = prop x:set x=$  
  end  
define cc = new c  
<cc>set x = 5 % ok - sets the field value to 5  
<cc>x % returns 5
```

The property `x` shadows the field `x`. This means that the field `x` can only be accessed directly except through the property.

The PROP language is based on the OBJ language with the addition of reference semantics and *properties*, as we proceed to describe.

Language PROP (continued)

Here are the grammar rules for defining properties in a class definition:

```
<props> ::= _PROPERTY_<VAR>_EQUALS_<prop>  
Prop (List<Token>_varList, _List<Prop>_propList)  
<prop> ::= _PROP_<exp>getExp_COLON_<exp>setExp  
Prop (Exp_getExp, _Exp_setExp)
```

When a variable bound to a property is evaluated, its `getExp` code is evaluated using the environment captured where the property is defined, and the value of the variable is the result of evaluating the `getExp` expression.

When a variable bound to a property is assigned to in a `set` expression, the expression is evaluated in the current environment. The environment where the property is defined is then extended by binding the special symbol `'_'` to the value of the RHS. The property's `setExp` expression is then evaluated in this extended environment, and the resulting value is the value of the `setExp` expression.

Language PROP (continued)

If a variable `z` [for example] is bound to a property whose `set` expression has side-effects [such as `nil`], an expression such as `set_z = ...` does nothing – including `z`.

```
define c =
  class
    field x
    method init = proc(x) {set <this>x = x ; self}
    property y = prop x:set x=$
    property z = prop x:nil
  end
define o = .<new c>init(3)
<o>[x,y,z]    % [3,3,3]
<o>set x = 5
<o>[x,y,z]    % [5,5,5]
<o>set y = 11
<o>[x,y,z]    % [11,11,11]
<o>set z = 42
<o>[x,y,z]    % [11,11,11]
```

Language PROP (continued)

We implement properties in the same way we implement call-by-name. Evaluating a property expression evaluates to a thunk-like object called a `PropRef` (which extends `Thunk`). This object captures the environment in which the property is defined and the property's `get` and `set` expressions.

The expressed value of a variable bound to a property is the value of evaluating the property's `get` expression in the captured environment. The thunk's `deRef` method.

Similarly, when assigning a value to a variable bound to a property, the `setRef` method is called, with the assigned value bound to the symbol '\$' and returning the value of the property's `set` expression.

Language PROP (continued)

If a field named `x` in a class has a property also named `x`, referring to of this object uses the `PropRef` instead of the variable. Consider the

```
define c =
  class
    static p = proc(t) set t=add1(t)
    field x
    method init = proc(x) {set <this>x=x ;
    property x = prop x : set x=$
    property y = prop x : set x=+($,$)
  end
define o = .<new c>init(3)
<o>{.p(x) ; x} % evaluates to 4
<o>{.p(y) ; x} % evaluates to 10
```

In the expression `.p(x)`, `x` refers to the property `x`. Since we are reference parameter passing semantics, the variable `t` is bound to and the expression `set_t=add1(t)` is evaluated using this binding that the RHS parts of property definitions are all evaluated in the environment includes only statics, fields, and methods, but not properties.

Language PROP (continued)

So far, a property is only defined in the context of a class definition, a role in object instantiation. It turns out that the behavior of properties can be useful even outside of the context of an object, especially to manage variables defined in a `let` expression. To make this explicit, we create a construct that has the following concrete syntax and abstract class structure:

<code><exp>:LetpropExp</code>	<code>::=_LETPROP_<letpropDecls>_IN_<exp></code>
	<div style="border: 1px solid black; padding: 2px;"><code>LetpropExp(LetpropDecls_letpropDecls,_Exp_exp)</code></div>
<code><letpropDecls></code>	<code>**=_<VAR>_EQUALS_<prop></code>
	<div style="border: 1px solid black; padding: 2px;"><code>LetpropDecls(List<Token>_varList,_List<Prop>_propList)</code></div>

Consider this (somewhat strange) example:

```
let
  x = 3
in
  letprop
    x = prop x : set x = 5
  in
    {set x = 42 ; x} % => 5
```

This expression evaluates to 5.

Language PROP (continued)

The `set` part of a `prop` is optional. If the `set` part is omitted, any use of the `set` operator to the variable results in a runtime exception. In this way we can implement “read-only” properties.

```
let
  x = 3
in
  letprop
    x = prop x %% no 'set' part, so x is read only
  in
    {set x = 42 ; x} % => runtime exception
```

Language PROP (continued)

It turns out that a read-only `prop` behaves just like a thunk in call-by-value (which, as you may recall, is also read-only), so we have the benefit of call-by-value when we want it!

```
let
  while = proc(test?, do, ans)
    letrec
      loop = proc()
        if test? then {do ; .loop()} else ans
    in
      .loop()
  sum = 0
  count = 10
  i = 1
in
  letprop
    test? = prop count
    do = prop { set sum = +(sum, i)
                ; set i = add1(i)
                ; set count = sub1(count)
              }
    ans = prop sum
  in
    .while(test?, do, ans) % => 55 = 1+2+...+10
```